
ВВЕДЕНИЕ 

 В настоящей работе выполнена реализация автомата – банкомата. 

 Для реализации автомата существует 2 способа: аппаратный и программный. 

Программная реализация выполняется с помощью языков программирования 

высокого уровня. Аппаратная реализация подразумевает построение устройств с 

памятью для запоминания текущего состояния. 

 Цель работы – закрепить навыки реализации конечных цифровых автоматов. 

Для выполнения поставленной цели необходимо выполнить следующие задачи: 

 Изучить задание в соответствии с вариантом; 

 Описать автомат, соответствующий условию задачи; 

 Изучить способы реализации автомата; 

 Выбрать один из изученных способов; 

 Программно реализовать описанный цифровой автомат; 

 Протестировать работу программной реализации; 

 Оформить отчёт. 

 

 В настоящей работе автомат реализован программным способом, так как этот 

способ наиболее прост в проектировании, отладке и тестировании автомата.  

 Задание (вариант 7): Автомат-банкомат. 

АВТОМАТ – БАНКОМАТ 

 Исходя из задания, автомат должен реализовать взаимодействие с 

пользователем – клиентом банка и выполнить следующие функции: 

 Идентификация счёта по информации с банковской карты; 

o Приняв карту, автомат должен определить номер банковского счёта, 

привязанного к этой карте. 

 Авторизация клиента банка по пин-коду; 

o Автомат даёт пользователю три попытки ввода четырёхзначного пин-

кода, и, в случае, если пин-код верный, предоставляет доступ к меню 

клиента. 

 Блокировка банковской карты; 

o В случае, когда пользователь совершил 3 неудачные попытки 

авторизоваться, автомат блокирует доступ к своим функциям и 

завершает работу. 

 Определение баланса счёта; 

o Автомат определяет и выводит на экран баланс счёта пользователя. 

 Пополнение баланса счёта; 

o Автомат открывает лоток для наличных и ожидает внесение средств. 

После внесения автомат закрывает лоток и обновляет баланс счёта 

пользователя. 

 Снятие наличных со счёта; 

o Автомат спрашивает у пользователя сумму снятия. В случае, если 

указанная сумма не превышает баланс счёта, автомат выдает 

указанную сумму наличными, в ином случае автомат выдает 

сообщение об ошибке. 

 Выдача чека; 

o Автомат спрашивает у пользователя, нужно ли печатать чек о снятии 

наличных средств. В случае положительного ответа автомат печатает 

чек. 

 Спецификация реализуемого автомата 

 Состояния автомата: 

 Q0 – Начальное состояние; 

 Q1 – Первая попытка авторизации; 

 Q2 – Вторая попытка авторизации; 

 Q3 – Третья попытка авторизации; 

 Q4 – Меню клиента; 

 Q5 – Ввод суммы списания; 

 Q6 – Выбор печати чека; 

 Q7 – Внесение наличных. 

 

 Входные сигналы: 

 a – Внесена банковская карта; 

 b – Введён верный пин-код; 

 с – Введён неверный пин-код; 

 d – Нажата кнопка «Определить баланс»; 

 e – Нажата кнопка «Снять наличные»; 

 f – Введена сумма, не превышающая баланс счёта; 

 g – Введена сумма, превышающая баланс счёта; 

 h – Нажата кнопка «Да»; 

 i – Нажата кнопка «Нет»; 

 j – Нажата кнопка «Пополнить счёт»; 

 k – Внесены наличные; 

 l – Нажата кнопка «Выход». 

 

 Выходные сигналы: 

 0 – Номер счёта; 

 1 -  Неудачная попытка авторизации; 

 2 – Успешная авторизация; 

 3 - Текущий баланс счёта; 

 4 – Приглашение ко вводу суммы; 

 5 – Сообщение «Недостаточно средств»; 

 6 – Выдача наличных; 

 7 – Печать чека; 

 8 – Открывание лотка; 

 9 – Закрывание лотка; 

 10 – Возврат банковской карты; 

 11 – Блокировка банковской карты; 

 Полученный цифровой автомат 

 На рисунке 1 изображен граф, описывающий автомат-банкомат в 

соответствии с данными, приведёнными в спецификации. 

Q0

Q1 Q2 Q3

Q4

Q5 Q6

Q7

a/0

c/1 c/1

c/11

b/2
b/2 b/2

d/3

l/10

e/4

g/5

f/6

h/7

i/-

j/8

k/9

 

Рисунок 1 - Автомат, заданный графом 

 Ниже представлена таблица 1, описывающая данный автомат. 

 

 

Таблица 1 - Таблица переходов и выходов автомата 

Сост.\Вх. 

сиг a b c d e f g h i j k l a b c d e f g h i j k l 

Q0 Q1            0            

Q1  Q4 Q2           2 1          

Q2  Q4 Q3           2 1          

Q3  Q4 Q0           2 11          

Q4    Q4 Q5     Q7  Q0    3 4     8  10 

Q5      Q6 Q4           6 5      

Q6        Q4 Q4           7 -    

Q7           Q4            9  

 

 

 

 Программная реализация полученного автомата 

 Для реализации полученного ранее цифрового автомата была разработана 

программа на языке Turbo Pascal. Ниже приведён листинг разработанной 

программы. 

 

program Project24; 

 

{$APPTYPE CONSOLE} 

 

{$R *.res} 

 

uses 

  System.SysUtils; 

 

 

 

Procedure Avtomat(); 

const sost_table: array[0..7, 'a'..'l'] of integer = ( 

(1, 8, 8, 8, 8, 8, 8, 8, 8, 8, 8, 8), 

(8, 4, 2, 8, 8, 8, 8, 8, 8, 8, 8, 8), 

(8, 4, 3, 8, 8, 8, 8, 8, 8, 8, 8, 8), 

(8, 4, 8, 8, 8, 8, 8, 8, 8, 8, 8, 8), 

(8, 8, 8, 4, 5, 8, 8, 8, 8, 7, 8, 0), 

(8, 8, 8, 8, 8, 6, 4, 8, 8, 8, 8, 8), 

(8, 8, 8, 8, 8, 8, 8, 4, 4, 8, 8, 8), 

(8, 8, 8, 8, 8, 8, 8, 8, 8, 8, 4, 8)); 

 

const out_table: array[0..7, 'a'..'l'] of integer = ( 

(0, 12, 12, 12, 12, 12, 12, 12, 12, 12, 12, 12), 

(12, 2, 1, 12, 12, 12, 12, 12, 12, 12, 12, 12), 

(12, 2, 1, 12, 12, 12, 12, 12, 12, 12, 12, 12), 

(12, 2, 11, 12, 12, 12, 12, 12, 12, 12, 12, 12), 

(12, 12, 12, 3, 4, 12, 12, 12, 12, 8, 12, 10), 

(12, 12, 12, 12, 12, 6, 5, 12, 12, 12, 12, 12), 

(12, 12, 12, 12, 12, 12, 12, 7, 12, 12, 12, 12), 

(12, 12, 12, 12, 12, 12, 12, 12, 12, 12, 9, 12)); 

 

const pin: shortstring = '1234'; 

 

 

Var sost: integer; 

    input: char; 

    option: char; 

    str: shortstring; 

    balance: integer; 

    sum: integer; 

    output: integer; 

 

begin 

    sost := 0; 

    balance := 10000; 

 

    repeat 

      case sost of 

        0: 

          begin 

            writeln('Нажмите Enter, чтобы вставить карту'); 

            readln; 

            input := 'a'; 

          end; 

        1..3: 

          begin 

            write('Введите пин-код: '); 

            readln(str); 

            if str='1234' then input := 'b' 

            else input := 'c'; 

          end; 

        4: 

          begin 

            writeln('1 - Узнать баланс'); 

            writeln('2 - Снять наличные'); 

            writeln('3 - Пополнить баланс'); 

            writeln('0 - Завершить работу'); 

            write('Введите число: '); 

            readln(option); 

            case option of 

              '1': input := 'd'; 

              '2': input := 'e'; 

              '3': input := 'j'; 

              else input := 'l'; 

            end; 

          end; 

        5: 

          begin 

            write('Введите сумму снятия: '); 

            readln(sum); 

            if (sum > balance) then input:= 'g' 

            else input := 'f'; 

          end; 

        6: 

          begin 

            writeln('Перчатать чек?'); 

            writeln('1 - Да'); 

            writeln('2 - Нет'); 

            write('Введите число: '); 

            readln(option); 

            if option = '1' then input := 'h' 

            else input := 'i'; 

          end; 

        7: 

          begin 

            write('Введите вносимую сумму: '); 

            readln(sum); 

            input := 'k'; 

          end; 

      end; 

 

      output := out_table[sost, input]; 

      case output of 

        0: writeln('Номер вашего лицевого счёта: 123456'); 

        1: writeln('Пин-код неверный!'); 

        2: writeln('Авторизация прошла успешно'); 

        3: writeln('Текущий баланс: ', balance); 

        4: writeln('Ввод суммы'); 

        5: writeln('На счету недостаточно средств'); 

        6: 

          begin 

            writeln('Автомат выдаёт наличные, ', sum); 

            balance := balance - sum; 

          end; 

        7: writeln('*Выдача чека*'); 

        8: writeln('Лоток открылся'); 

        9: 

          begin 

            writeln('Лоток закрылся'); 

            balance := balance + sum; 

          end; 

        10: writeln('Банкомат возвращает банковскую карту'); 

        11: writeln('КАРТА ЗАБЛОКИРОВАНА!'); 

      end; 

 

      sost := sost_table[sost, input]; 

    until sost=0; 

 

 

end; 

 

begin 

 

Avtomat; 

readln; 

 

end. 

 На рисунке 2 показан пример работы программы. 

 

Рисунок 2 - Пример работы программы 

 Тестирование цифрового автомата 

 В таблице 2 приведены наборы тестов для полученного цифрового автомата. 

Входная 

последовательность 
Ожидаемый результат Полученный результат 

Трижды введён 

неверный пин-код 
Блокировка карты Блокировка карты 

Попытка снять сумму, 

превышающую баланс 

счёта 

Сообщение об ошибке Сообщение об ошибке 

Пополнение баланса 

счёта 

Увеличение числа 

денежных средств на 

счету 

Увеличение числа 

денежных средств на 

счету 

Авторизация 
Предоставление меню 

клиента 

Предоставление меню 

клиента 

Определение баланса Вывод текущего баланса Вывод текущего баланса 

Снятие наличных 
Автомат выдает 

наличные 

Автомат выдает 

наличные 

Печать чека Автомат выдаёт чек Автомат выдаёт чек 

 

  

 

 Примеры работы конечного автомата на тестовых случаях приведены на 

рисунках 3-9. 

 

Рисунок 3 - Блокировка карты 

Рисунок 4 - Сообщение об ошибке 

Рисунок 5 - Пополнение баланса 

 

 

Рисунок 6 - Авторизация 

Рисунок 7 - Определение баланса 

  

 

 

Рисунок 8 - Снятие наличных 

 

 

Рисунок 9 - Печать чека 

ЗАКЛЮЧЕНИЕ 

1) В результате выполнения домашнего задания был изучен способ 

программной реализации конечных цифровых автоматов 

2) В ходе программной реализации автомата была разработана программа, 

реализующая цифровой автомат в среде разработки Turbo Pascal 

3) Закреплены навыки подготовки и оформления отчета по проделанной 

работе с учетом требований ГОСТ 7.32-2017. 

